query: PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
  
  SELECT DISTINCT ?item ?itemLabel WHERE {
    ?item wdt:P31 wd:Q5.
    ?item (wdt:P106/wdt:P279*) wd:Q36180.
    { ?item wdt:P27 wd:Q29999. }
    UNION
    {
      ?item wdt:P19 ?pob.
      ?pob wdt:P17 wd:Q55.
    }
    UNION
    {
      ?item wdt:P937 ?pow.
      ?pow wdt:P17 wd:Q55.
    }
    UNION
    {
      ?item wdt:P551 ?por.
      ?por wdt:P17 wd:Q55.
    }
    UNION
    {
      ?item wdt:P20 ?pod.
      ?pod wdt:P17 wd:Q55.
    }
    ?item wdt:P570 ?death.
    SERVICE wikibase:label { bd:serviceParam wikibase:language "nl, en". }
    FILTER(?death >= "1949-01-01T00:00:00Z"^^xsd:dateTime)
    FILTER(?death < "1950-01-01T00:00:00Z"^^xsd:dateTime)
  }